



3.2. Shock 2 Human consumption

Jörg Rieger
Thünen Institute

Scenarios for changes in food consumption become more important

- Recent CAPRI studies:

Global Food Security 28 (2021) 100437

Contents lists available at ScienceDirect

Global Food Security

journal homepage: www.elsevier.com/locate/gfs



Paying the price for environmentally sustainable and healthy EU diets

Catharina Latka^{a,*}, Marijke Kuiper^b, Stefan Frank^c, Thomas Heckelei^a, Petr Havlík^c, Heinz-Peter Witzke^a, Adrian Leip^d, Hao David Cui^b, Anneleen Kuijsten^{e,f}, Johanna M. Geleijnse^{e,f}, Michiel van Dijk^{b,c}



nature communications

Article <https://doi.org/10.1038/s41467-023-41789-3>

The global and regional air quality impacts of dietary change

Received: 8 March 2022 Accepted: 18 September 2023

Marco Springmann^{1,2}, Rita Van Dingenen³, Toon Vandyck^{4,5}, Catharina Latka⁶, Peter Witzke^{6,7} & Adrian Leip⁸



STUDY



Agriculture, forestry and food in a climate neutral EU

The land use sectors as part of a sustainable food system and bioeconomy



Received: 4 July 2022 | Revised: 29 November 2022 | Accepted: 19 January 2023

DOI: 10.1111/1477-9552.12530



ORIGINAL ARTICLE

From fork to farm: Impacts of more sustainable diets in the EU-27 on the agricultural sector

Jörg Rieger¹ | Florian Freund² | Frank Offermann¹ | Inna Geibel² | Alexander Gocht¹



Theoretical Background: Shock Human consumption




Human consumption in CAPRI

Generalized Leontief Expenditure System

$$F = \sum_i D_i P_i$$

$$G = \sum_i \sum_j B_{i,j} \sqrt{P_i P_j}$$



$$\frac{\partial F}{\partial P_i} = F_i = D_i$$

$$\frac{\partial G}{\partial P_i} = G_i = \sum_j B_{i,j} \sqrt{P_j / P_i}$$

Expenditure remaining after commitments are covered

$$\Rightarrow x_i = \left[D_i + \frac{G_i}{G} (Y - F) \right] Pop$$

Value of minimum commitments

D_i = Consumption independent of prices and income

Human consumption in CAPRI

"\gams\arm\market_model.gms"

```

* ----- Behavioural functions
*$onlisting
*
* --- definition of F as sum of Di multiplied with prices for the Generalised Leontief expenditure funtion
*         PD: commitments / linear terms in the individual demand functions
*         (in kg per capita, therefore prices are expressed in Euro/kg <=> CPRI [Euro/ton] * 0.001)
*
GLDemandFS_(RMS) $ sum(xxx $ v_consQuant.range(RMS,XXX),1) ..
*
  v_GLDemandFS(RMS) /
*
  --- the following term (also found o the RHS) is only for scaling purposes
  to bring the equation around unity
*
  (SUM(XX1 $ p_pdGL(RMS,XX1,"CUR"),
    DATA(RMS,"CPRI",XX1,"CUR") * p_pdGL(RMS,XX1,"CUR")*1.E-3) + 0.1)
*
  =E= SUM(XX1 $ p_pdGL(RMS,XX1,"CUR"),
    v_consPrice(RMS,XX1) * p_pdGL(RMS,XX1,"CUR")*1.E-3)
  / (SUM(XX1 $ p_pdGL(RMS,XX1,"CUR"),
    DATA(RMS,"CPRI",XX1,"CUR") * p_pdGL(RMS,XX1,"CUR")*1.E-3) + 0.1);
*
* --- definition of function G for the Generalised Leontief expenditure funtion
*         (per capita)
*
GLDemandGS_(RMS) $ sum(xxx $ v_consQuant.range(RMS,XXX),1) ..
*
  --- due to DF in the calibration of the parameter set, it is possible
  to fix v_GLDemandGS durign a value. That value is 10000. and
  use in the equation for scaling to bring the equation to unity in the
  calibration point
*
  v_GLDemandGS(RMS)/10000. =E= SUM( (XX1,YY1) $ p_pbGL(RMS,XX1,YY1,"CUR"),
    p_pbGL(RMS,XX1,YY1,"CUR")
    * SQRT(v_consPrice(RMS,XX1)*v_consPrice(RMS,YY1)*1.E-6) )/10000.;

```

$$F = \sum_i D_i P_i$$

D_i = Consumption independent of prices and income

$$G = \sum_i \sum_j B_{i,j} \sqrt{P_i P_j}$$

Human consumption in CAPRI

"\gams\arm\market_model.gms"

```

* --- definition of first derivatives of G called Gi for the Generalised Leontief expenditure funtion
*   (per capita)
*
GLDemandGis_(RMS,XXX) $ ( (v_consQuant.LO(RMS,XXX) ne v_consQuant.UP(RMS,XXX))
    $ DATA(RMS,"HCon",XXX,"CUR") ..


v_GLDemandGis(RMS,XXX)

*   --- the following is only a scaling term to bring the equation around unity
*   in calibration point. It is also found on the RHS
*
/ (SUM( YY1 $ p_pbG1(RMS,XXX,YY1,"CUR"),
    p_pbGL(RMS,XXX,YY1,"CUR")
    * SQRT(DATA(RMS,"CPRI",YY1,"CUR")/DATA(RMS,"CPRI",XXX,"CUR"))) + 0.1)

=E= [SUM( YY1 $ p_pbG1(RMS,XXX,YY1,"CUR"),
    p_pbGL(RMS,XXX,YY1,"CUR")
    * SQRT(v_consPrice(RMS,YY1)/DATA(RMS,"CPRI",XXX,"CUR"))
    * SQRT(DATA(RMS,"CPRI",XXX,"CUR")/v_consPrice(RMS,XXX))
    ]

*   ---- the division / multiplication by DATA(RMS,"CPRI",XXX,"CUR")
*   cancel out and are only introduced to stabilize
*   the numerical solution
*
/ (SUM( YY1 $ p_pbG1(RMS,XXX,YY1,"CUR"),
    p_pbGL(RMS,XXX,YY1,"CUR")
    * SQRT(DATA(RMS,"CPRI",YY1,"CUR")/DATA(RMS,"CPRI",XXX,"CUR"))) + 0.1);

```



Human consumption in CAPRI

"\gams\arm\market_model.gms"

```

* ----- definition of human consumption for the Generalised Leontief expenditure funtion
*
XiS_(RMS,XXX) $ ( (v_consQuant.LO(RMS,XXX) ne v_consQuant.UP(RMS,XXX))
                  $ DATA(RMS,"HCon",XXX,"CUR")) ..
*
( v_consQuant(RMS,XXX) $ ( v_consQuant.LO(RMS,XXX) GT 0)
+ v_consQuantNeg(RMS,XXX) $ ( v_consQuant.LO(RMS,XXX) LE 0)
)
/(DATA(RMS,"HCon",XXX,"CUR")+0.1) =E=
*
((v_GLDemandGis(RMS,XXX)/v_GLDemandGS(RMS)
 * (DATA(RMS,"Ince","Levl","CUR")/DATA(RMS,"INHA","LEVL","CUR") - v_GLDemandFS(RMS))
 + p_pdGL(RMS,XXX,"CUR")) * DATA(RMS,"INHA","LEVL","CUR")/1000.)/(DATA(RMS,"HCon",XXX,"CUR")+0.1);

```

$$\Rightarrow x_i = \left[D_i + \frac{G_i}{G} (Y - F) \right] Pop$$

Value of minimum commitments

D_i = Consumption independent of prices and income

Diet shifts in CAPRI

Implementation of diet shifts in CAPRI model

(e.g. 10% decrease in demand for beef)

- GL demand function shifted under constant price assumption
 - New demand curve through 10% less demand for beef at the same price
- Demand curve is shifted by adjusting the commitment terms („p_pdGL“)
 - Recalibration of commitment terms to match new demand at original price
 - Analogous to shifting the constant term in linear demand functions

$$PerCap_i = D_i + \frac{G_i}{G} (y - F); D_i = p_pdGL$$

Diet shifts in CAPRI

```
set scenshiftedCols(COLS) / UVAG
$IF %MARKET_M%==ON          TCost
$IF %MARKET_M%==ON          HCON
$IF %MARKET_M%==ON          FEED
$IF %MARKET_M%==ON          PROD
$IF %MARKET_M%==ON          PROC ,
$IF %MARKET_M%==ON          SET.bioScenShifted
$IF %MARKET_M%==ON          SET.foreOIndArtif
$IF %MARKET_M%==ON          LEVEL
```

```
$IFI %MARKET_M%==ON SET ScenItems(COLS) "Items which can be trend shifted via scendefinition" / set.scenshiftedCols /;
```

```
DATA(RMS,"HCON","beef","PercentageChange") = -10;
```

arm\prep_market.gms

```
*
* --- shift human consumption, feed, processing, production according to exogenously defined
*      (using XX1 permits to pick up shifters for PROD.LAND = land area for non-supply model
*
DATA(RMS,ScenItems,XX1,"CUR") $ DATA(RMS,ScenItems,XX1,"AbsoluteLevel")
= DATA(RMS,ScenItems,XX1,"AbsoluteLevel");
*
DATA(RMS,ScenItems,XX1,"CUR") $ DATA(RMS,ScenItems,XX1,"AbsoluteChange")
= DATA(RMS,ScenItems,XX1,"CUR") + DATA(RMS,ScenItems,XX1,"AbsoluteChange");
*
DATA(RMS,ScenItems,XX1,"CUR") $ DATA(RMS,ScenItems,XX1,"ChangeFactor")
= DATA(RMS,ScenItems,XX1,"CUR") * DATA(RMS,ScenItems,XX1,"ChangeFactor");
*
DATA(RMS,ScenItems,XX1,"CUR") $ DATA(RMS,ScenItems,XX1,"PercentageChange")
= DATA(RMS,ScenItems,XX1,"CUR") * (1. + DATA(RMS,ScenItems,XX1,"PercentageChange")/100);
```

```
v_consQuant.L(RMS,XX) = DATA(RMS,"HCon",XX,"CUR");
```

Diet shifts in CAPRI

\gams\arm\trim_gl_commits.gms

GAMS file : TRIM_GL_COMMITS.GMS

@purpose : Recalibrate commitments of Generalized Leontief Demand System to given prices and quantities by minimizing differences to given commitments

```
EQUATIONS trimGLCommits1_ "Value of commitments"
           trimGLCommits2_ "Demand equation"
           trimGLCommits3_ "Definition of squared differences between given and calibrated commitments";
```

```
VARIABLES v_commitPar(RMS,XX1) "Commitment parameter of Generalized Leontief Demand System"
           v_Obje "Objective function";
```

```
*
trimGLCommits1_(RMS) ..
*
v_GLDemandFS(RMS) =E= SUM(XX1 $ ((DATA(RMS,"Hcon",XX1,"CUR") or SAMEAS(XX1,"INPE")) and p_pdGL(RMS,XX1,"CUR")),
                          v_consPrice.L(RMS,XX1) * v_commitPar(RMS,XX1)*1.E-3);
```

```
trimGLCommits2_(RMS,XX) $ (DATA(RMS,"Hcon",XX,"CUR") and v_GLDemandGis.L(RMS,XX) and p_pdGL(RMS,XX,"CUR")) ..
```

```
v_consQuant.L(RMS,XX) =E= (v_GLDemandGis.L(RMS,XX)/v_GLDemandGs.L(RMS)
                          * ( DATA(RMS,"Ince","Levl","CUR")/DATA(RMS,"INHA","LEVL","CUR")
                              - v_GLDemandFS(RMS)) + v_commitPar(RMS,XX)
                          * DATA(RMS,"INHA","LEVL","CUR")/1000.);
```

$$x_i = [D_i + \frac{G_i}{G}(y - F)] * Pop$$

```
*
trimGLCommits3_ .. v_Obje =E= SUM( (RMS,XX1) $ (v_GLDemandGis.L(RMS,XX1) and p_pdGL(RMS,XX1,"CUR")),
                                  SQR(v_commitPar(RMS,XX1) - P_pdGL(RMS,XX1,"CUR")));
```

```
MODEL m_trimGLCommits "Model to calibrate commitments of GL demand system to given prices and quantities"
/ trimGLCommits1_,trimGLCommits2_,trimGLCommits3_ /;
```

*

Diet shifts in CAPRI

```
\gams\arm\recalibrate_commitments.gms
```

```
@purpose : Redefine commitments terms of Generalized Leontief demand  
System to match given quantities and prices
```

```
m_trimGLCommits.Solprint = 1;  
m_trimGLCommits.iterlim = 10000;  
m_trimGLCommits.SolveLink = 5;  
SOLVE m_trimGLCommits USING NLP Minimizing v_obje;  
if ( EXECERROR > 0, abort "internal error");  
  
if(m_trimGLCommits.numinfes > 0,  
  execute_unload "%resdir%\chk_p_pdGL.gdx";  
  abort "m_trimGLCommits.numinfes > 0 ", m_trimGLCommits.numinfes;  
);  
  
p_pdGL(RMS,XX1,"CUR") = v_commitPar.L(RMS,XX1);
```

Diet shifts in CAPRI

Consumption-Related Indicators in CAPRI

HCON = Food intake + market losses (**LOSM**) + industrial use (**INDM**) + consumption-stage losses (**LOSC**)

HCOM = Food intake + consumption-stage losses (**LOSC**)

N_CAL = Food intake (kcal per capita and day)

N_PRO = Food intake (g proteins per capita and day)

N_FAT = Food intake (g fat per capita and day)

.....

INHA = Food intake (kg per capita and year)

➤ *dataOut(DE000000, "", INHA, BEEF, 2030)*

Diet shifts in CAPRI

Nutrition details [0]

- Meta >
- Scenario shifter >
- CAP >
- Policy indicators >
- Food**
- Markets** >
 - Product balances, detailed
- BioFuels > Demand Balances
- Trade > Product balances market model
- Farm > Market balances without intra trade
- Farm - totals > Yields, levels and production
- Farm EU > Global irrigation
- Fertilizer > Global land use and land aggregates
- Environment > Product Balances
- Multi-Functionality >
 - Food consumption
 - Consumption/Losses/Industrial use in food chain
 - Nutrition details**
- Energy >
- Feed >
- Welfare > Market model balances - decomposition
- Prices > Milk fat and protein
- Pesticides > Milk products

Nutrition details [0]

Region
Germany

Food intake (kg per capita and day after losses)	0.70
Energ_cal_per_capita_and_day	7.42
Protein_g_per_capita_and_day	0.32
Lipid_Tot_g_per_capita_and_day	0.13
Fiber_TD_g_per_capita_and_day	0.20
Sugar_Tot_g_per_capita_and_day	0.02
Calcium_mg_per_capita_and_day	1.03
Iron_mg_per_capita_and_day	0.09
Magnesium_mg_per_capita_and_day	3.37
Potassium_mg_per_capita_and_day	8.18
Sodium_mg_per_capita_and_day	0.04
Zinc_mg_per_capita_and_day	0.08
Selenium_microg_per_capita_and_day	0.04
Vit_C_mg_per_capita_and_day	
Thiamin_VIT_B1_mg_per_capita_and_day	0.01
Riboflavin_VIT_B2_mg_per_capita_and_day	0.00
Vit_B6_mg_per_capita_and_day	0.00
Folate_Tot_microg_per_capita_and_day	1.07
Vit_B12_microg_per_capita_and_day	
Vit_A_International_Unit_per_capita_and_day	
Vit_E_mg_per_capita_and_day	0.02
Vit_D_microg_per_capita_and_day	
FA_Sat_g_per_capita_and_day	0.02
FA_Mono_g_per_capita_and_day	0.04
FA_Poly_g_per_capita_and_day	0.05

CAPRI-Scenario: Reduced meat consumption


Decrease in meat-based calorie consumption above a given threshold (minimum requirements in kcal for healthy diets)

- Threshold levels according to recommendations by Lancet Commission on healthy diets based on a diet of 2500 kcal per day
 - Calorie intake from animal products: 340 kcal/capita/day
 - Calorie intake from meat: 128 kcal/capita/day
 - **Calorie intake from beef: 7.5 kcal/capita/day**
- Reduction of meat consumption above some threshold level (“luxury consumption”) for the daily calorie requirements (kcal/day/capita)
- Partial compensation of reduced calorie intake from meat products e.g. 50% of the cut in meat calories with vegetables, fruits and legumes
 - Consideration of increasing preferences for vegetarian products

CAPRI-Scenario „Demand_shock_beef.gms“

- **Reduction** of **beef** consumption above some threshold level (“luxury consumption”) by **30%** in the **EU**
- **Compensation** of reduced calorie intake from beef with **vegetables, fruits and legumes** by **50%** in the EU

```
› gams › pol_input › userScens  
Demand_shock_beef.gms
```



CAPRI-Scenario „Demand_shock_beef.gms“

```
set X_T_foodagg(rows, foodagg)
  / (BEEF).beef
  (PULS, SOYA, TOMA, OVEG, APPL, OFRU, CITR, TAGR).leguveg
  /;
```

```
* -----
* Parameter p_dietChange to define reference diets and dietary shifts
* -----
parameter p_dietChange(EU27_Brexit, *, *, *) "parameter to define diet change scenarios" ;

execute_load "%results_in%\capmod\res_2_%BAS%%SIM%cap_after_2014_ref%reg_agg%.gdx" p_DataOutTemp=dataout;

* --- calories by food group in the reference scenario

p_dietChange(EU27_Brexit, "calories_ref", foodagg, "%SIMY%")
  = sum(X_T_foodagg(rows, foodagg), p_DataOutTemp(EU27_Brexit, "", "n_cal", rows, "%SIMY%"));
```

```
*Define the products for which the consumer demand will be reduced (e.g. redmeat or allmeat)
$setglobal cutmeat  beef
*Define the Cut in calories intake from meat in (% decrease in calories intake above threshold)
$setglobal cutrate_meat  30

*Define substitution products for reduced calorie intake from meat (e.g. legumes or leguveg)
$setglobal subst_veg  leguveg
* Define the rate of substitution (% compensation of the reduced meat calories )
$setglobal subst_rate_veg  50
```

```
parameter p_cutCons(rows, simyy) "relative cut for excess meat-based calorie consumption";
p_cutCons("%cutmeat%", "%SIMY%") = %cutrate_meat%/100;
```

```
parameter p_cutCons_v(rows, simyy) "substitution of relative cut for excess meat-based calorie consumption with vegetables and fruits";
p_cutCons_v("%subst_veg%", "%SIMY%") = %subst_rate_veg%/100;
```

CAPRI-Scenario „Demand_shock_beef.gms“

```
parameter p_threshCAL(rows)
  /beef      7.5/;
```

← Lancet recommendations (kcal per capita and day)

```
* --- meat calories after cut (but before endogenous adjustments in simulation) reduction on meat-based calories only above threshold level

p_dietChange(EU27_Brexit,"calories_sim","%cutmeat%","%SIMY%")
= p_dietChange(EU27_Brexit,"calories_ref","%cutmeat%","%SIMY%")
- max(0,min(p_dietChange(EU27_Brexit,"calories_ref","animpro","%SIMY%")-p_threshCAL("animpro"),
  p_cutCons("%cutmeat%","%SIMY%")*(p_dietChange(EU27_Brexit,"calories_ref","%cutmeat%","%SIMY%")-p_threshCAL("%cutmeat%"))
));

* --- define relative cut for meat

p_dietChange(EU27_Brexit,"PercentageChange","%cutmeat%","%SIMY%") $ p_dietChange(EU27_Brexit,"calories_ref","%cutmeat%","%SIMY%")
= (p_dietChange(EU27_Brexit,"calories_sim","%cutmeat%","%SIMY%")/p_dietChange(EU27_Brexit,"calories_ref","%cutmeat%","%SIMY%")-1)*100;

* --- define relative shift in meat consumption

DATA(EU27_Brexit,"HCON","%cutmeat%","PercentageChange") = p_dietChange(EU27_Brexit,"PercentageChange","%cutmeat%","%SIMY%");
```

➤ Substitution of beef calories with fruits, vegetables and legumes

```
p_dietChange(EU27_Brexit,"PercentageChange","%subst_veg%","%SIMY%") $ p_dietChange(EU27_Brexit,"calories_ref","%subst_veg%","%SIMY%")
* ... increase in 50% of the absolute cut in meat calories
* = 0.5*(p_dietChange(EU27_Brexit,"calories_ref","%cutmeat%","%SIMY%")-p_dietChange(EU27_Brexit,"calories_sim","%cutmeat%","%SIMY%"))
= p_cutCons_v("%subst_veg%","%SIMY%")*(p_dietChange(EU27_Brexit,"calories_ref","%cutmeat%","%SIMY%")-p_dietChange(EU27_Brexit,"calories_sim","%cutmeat%","%SIMY%"))
* ...relative to the reference run consumption of legumes&fruits&vegetables
/ p_dietChange(EU27_Brexit,"calories_ref","%subst_veg%","%SIMY%")*100;

* --- define relative shift in legumes and fruits&vegetables consumption

DATA(EU27_Brexit,"HCON",xleguveg,"PercentageChange") = p_dietChange(EU27_Brexit,"PercentageChange","%subst_veg%","%SIMY%");
```

CAPRI-Scenario „Demand_shock_beef.gms“

The screenshot displays the CAPRI TS 2024 software interface. The main window is titled "CAPRI General settings" and is divided into several sections:

- General settings:** Includes tabs for "General settings", "Modules and algorithm", "Reporting", and "Debug options".
- CAPRI worksteps:** A list of tasks with radio buttons:
 - Installation
 - Build database
 - Generate baseline
 - Run scenario
 - Tests and Reporting
- CAPRI tasks:** A list of tasks with radio buttons:
 - Define scenario
 - Run scenario with market model
 - Run scenario without market model
 - Test alternative market model
 - Run scenario only with market model
- Scenario description:** Fields for "Dir:" (userScens), "Files:" (Demand_shock_beef), "Aggregation file:" (defaultA), and "Scenario group:" (NoGroup).
- Years:** A table showing simulation years from 2004 to 2085. The "Base year" is set to 2017. The "Last simulation year" is set to 2050.
- Regions:** A list of countries including BL000000, DK000000, DE000000, EL000000, ES000000, FR000000, IR000000, IT000000, NL000000, and AT000000. The "Regional breakdown" is set to "NUTS2".
- FAO options:** "FAOSTAT updated to 2021" is checked, and "Non-default FAO trade matrix vintage" is unchecked.
- Buttons:** "Compile GAMS", "Start GAMS", "Stop GAMS", "Hide/Unhide controls", and "Exploit results" (highlighted with a red box).

CAPRI-Scenario „Demand_shock_beef.gms“

Result exploitation

Regional Aggregation
defaultA
MajorPorkProd

Country selection
EU BL DK DE
EL ES FR IR
IT NL AT PT
SE FI UK CZ
HU PL SI SK
EE LT LV CY
MT BG RO NO
TR TURAL MK
CS MO HR BA
KO

Regional level 029

Base year selection 0408101217

Simulation year selection
000102030405060708
091011121314151617
181920212223242526
272829303132333435
363738394041424344
454647484950515253
545556575859606162
636465666768697071
727374757677787980
8182838485

Scenario 1 res_2_1730cap_after_2014_refdefaultA

Scenario 2 res_2_1730userScens_Demand_shock_beefdefaultA

Scenario 3

Scenario 4

Scenario 5

Scenario 6

Scenario 7

Scenario 8

Scenario 9

Scenario 10

Scenario 11

Scenario 12

Scenario 13

Scenario 14

Scenario 15

Scenario 16

Select scenarios

Expectations for European agricultural sector

- Supply
- Prices
- Imports/Exports
- Nutrient balances (N-Surplus)
- Global warming potential

Exercise

1. The producer **price for beef** in the EU-27 changes by _____%
2. The country with the highest reduction in **beef meat activities** (number of animals), in **absolute terms**, is _____
3. The **beef supply** (1000t) in the EU-27 changes by _____%
4. The change in the **supply of fodder** (1000t) for the EU-27 is _____%
5. For the **EU beef exports** change by _____% and **imports** by _____% (*Hint: "market balances without intra trade"*)
6. For the **EU exports of vegetables and permanent crops** change by _____% and **imports** by _____%
7. Which are the **4 countries** in the **EU-27** with the **highest reduction in global warming potential from agriculture** in total terms
 1. _____
 2. _____
 3. _____
 4. _____

(*Hint: "Environment => Environmental indicators"*)
8. How high are the absolute changes in net emissions (1000t) (Total global warming potential) in the EU _____ Non-EU _____ and World _____

(*Hint: "Emissions by commodity" => "Emissions for emission leakage"*)

Recent CAPRI studies

Rieger, J., Freund, F., Offermann, F., Geibel, I. & Gocht, A. (2023) From fork to farm: Impacts of more sustainable diets in the EU-27 on the agricultural sector. *Journal of Agricultural Economics*, 74, 764–784.

<https://doi.org/10.1111/1477-9552.12530>

Himics, M., Giannakis, E., Kushta, J., Hristov, J., Sahoo, A. and Perez-Dominguez, I., 2022. Co-benefits of a flexitarian diet for air quality and human health in Europe. *Ecological Economics*, 191, p.107232.

<https://doi.org/10.1016/j.ecolecon.2021.107232>